Representations of A_5 using MAGMA

David Joyner

9-19-2000

Abstract

We show how to use some commands in MAGMA [MAGMA] to investigate finite group representations. In particular, we study the representations of A_5. This note is written for the first-time user of MAGMA but assumes that the reader knows basic things about groups and representations.

Contents

1 Introduction

Very briefly, in this note we shall first learn how to enter a finite group, how to list their elements, their conjugacy classes, and their centralizers. Then we shall study their representations, more precisely their characters [1]. We shall also study the induced representations $ind_H^G\ \chi$, where H is a certain subgroup of G, and G is either A_5 or its 2-fold cover $SL(2,5)$. We shall discuss A_5 from the following points of view:

- as a permutation group,
- as a matrix group ($SL(2,4)$),
- as the quotient of a matrix group ($PSL(2,5)$),
- as a finitely presented group ($<\ a, b \mid a^2 = b^3 = (ab)^5 = 1\ >$).

MAGMA [MAGMA] is a non-free[2] software package "designed to solve computationally hard problems in algebra, number theory, geometry and combinatorics". The name "magma" is explained by another quote from the MAGMA documentation: "The primary concept in the design of the Magma system is that of a 'magma'. Following Bourbaki, a magma can be defined as a set with a law of composition." The MAGMA project has been headed by John Cannon in the Department of Mathematics at the University of Sydney in Australia. It now has many contributors from all over the world (see the section "Acknowledgements" in the documentation). There is also a manual (both in latex and in html) and a nice introduction (in dvi and ps), both including many examples.

It will be assumed that you have MAGMA up and running. Sometimes a MAGMA command will be given but the MAGMA output will not be given (though quite often we give both), since it will be assumed you can see it yourself. Many commands assume that the commands previously discussed have already been entered.

Once you start MAGMA, you will see a banner like this:

[1]Until the last section, we shall assume all the word "representation" refers to a finite dimensional complex representation. The last section briefly discusses modular representations.

[2]Not free but not for profit either. The cost of the license is put into further developement of the package.

```
Magma V2.7-2      Mon Sep  4 2000 08:00:45 on alm      [Seed = 76461011]
Type ? for help.  Type <Ctrl>-D to quit.
```

From this you already know the date, that I am on a machine named `alm`, and that I have started MAGMA, version 2.7.2.

I recommend you log every session. To do this, type

```
> SetLogFile("/home/wdj/magma27/altern_gp.log");
```

for example (we will be doing an example based on the alternating group A_5, so the file name will be called `altern_gp.log`). You don't type " `>`", by the way, as that is simply the MAGMA cursor beginning each line. Also, be sure you save the log to a directory you have read-wite access to.

2 Entering groups

To begin, let us enter the alternating group into MAGMA by typing

```
>A5:=AlternatingGroup(5);
```

You can see by typing `#A5;` that it has 60 elements. Its elements are obtained by typing

```
>a5:=Set(A5);
```

Since $A_5 \cong PSL(2, \mathbb{F}_4)$, you can also enter this group into MAGMA by the command

```
>PSL24:=PSL(2,4);
```

if you prefer. This will also be a permutation group in MAGMA. On the other hand, if you enter $SL(2, \mathbb{F}_4)$ using

```
>SL24:=SL(2,4);
```

(which is the same group after all, since we're in characteristic 2), you will get a group of *matrices*, not a permutation group, whose elements are written in the form

```
[$.1^2      0]
[    0   $.1],

[$.1^2      0]
[$.1^2   $.1],
```

```
[$.1^2     0]
[  $.1   $.1],

[$.1^2     0]
[    1   $.1],
```

```
<stuff omitted>
```

You may also enter this group as

```
>PSL25:=PSL(2,5);
```

if you wish. It is also a permutation group in MAGMA and by typing

```
>Set(PSL25);
```

you find its elements are

```
{
    (1, 4, 6, 3, 2),
    (1, 4, 3, 2, 5),
    (1, 4, 2)(3, 6, 5),
    (1, 3, 4, 2, 6),
    (1, 3, 5)(2, 4, 6),
    (1, 5, 6, 2, 3),
    (1, 6, 2)(3, 4, 5),
    (1, 6, 4)(2, 5, 3),
    (1, 5)(3, 6),
    (1, 2)(5, 6),
    (2, 6)(3, 5),
    (1, 5, 4)(2, 6, 3),
    (2, 4)(3, 6),
    (2, 4, 6, 5, 3),
    (1, 3)(2, 5),
    Id(PSL25),
    (1, 2, 6, 4, 5),
    (1, 6, 5, 2, 4),
    (1, 5, 4, 6, 2),
    (1, 5, 2, 3, 4),
    (1, 2, 5, 3, 6),
    (1, 6)(2, 3),
    (1, 4)(2, 6),
    (1, 6, 3)(2, 5, 4),
    (1, 4, 5, 6, 3),
```

```
    (1, 4)(3, 5),
    (2, 6, 3, 4, 5),
    (1, 3, 4)(2, 5, 6),
    (1, 5, 3)(2, 6, 4),
    (1, 3, 2)(4, 5, 6),
    (1, 6, 4, 3, 5),
    (1, 2, 4)(3, 5, 6),
    (1, 2, 4, 5, 3),
    (1, 5, 2)(3, 6, 4),
    (2, 5)(4, 6),
    (1, 3, 2, 6, 5),
    (1, 3, 6, 5, 4),
    (2, 3)(4, 5),
    (1, 2, 3, 6, 4),
    (1, 5, 3, 4, 6),
    (1, 5)(2, 4),
    (1, 2, 6)(3, 5, 4),
    (1, 4, 5)(2, 3, 6),
    (1, 4, 2, 5, 6),
    (1, 3, 5, 4, 2),
    (1, 6, 2, 4, 3),
    (1, 3, 6)(2, 4, 5),
    (2, 5, 4, 3, 6),
    (2, 3, 5, 6, 4),
    (1, 3)(4, 6),
    (1, 6, 3, 5, 2),
    (1, 2, 5)(3, 4, 6),
    (1, 6, 5)(2, 3, 4),
    (1, 4, 3)(2, 6, 5),
    (1, 2, 3)(4, 6, 5),
    (1, 5, 6)(2, 4, 3),
    (1, 2)(3, 4),
    (1, 4, 6)(2, 3, 5),
    (1, 6)(4, 5),
    (3, 4)(5, 6)
}
```

All these groups are known to be isomorphic (probably a proof can be found in [R] but I'm not sure). One way to see this using MAGMA is to type

```
>IsSimple(A5); #A5;
>IsSimple(PSL25); #PSL25;
>IsSimple(SL24); #SL24;
```

They are all simple groups of size 60, so by the classification of simple groups they must be isomorphic!

Next, we compute all the conjugacy classes of A_5:

```
> A5_classes:=ConjugacyClasses(A5);
```

MAGMA will reply with something like

```
Conjugacy Classes of group A5
-----------------------------
[1]     Order 1       Length 1
        Rep Id(A5)

[2]     Order 2       Length 15
        Rep (1, 2)(3, 4)

[3]     Order 3       Length 20
        Rep (1, 2, 3)

[4]     Order 5       Length 12
        Rep (1, 2, 3, 4, 5)

[5]     Order 5       Length 12
        Rep (1, 3, 4, 5, 2)
```

This is a list of classes, sizes and representatives. To get a representative of an element x in A_5, you can use the `ClassRepresentative` command. For example, if $x = (1, 2, 4)$, type

```
ClassRepresentative(A5,A5!(1,2,4));
```

Here is a "coersion": to force MAGMA to recognize $(1, 2, 4)$ as an element of A_5 you preface it by juxtaposing `A5!` before it. Now let us find all the non-trivial centralizers in A_5, up to conjugacy. Type

```
> C1:=Centralizer(A5,A5!(1,2)(3,4)); C1;
Permutation group C1 acting on a set of cardinality 5
Order = 4 = 2^2
```

```
    (1, 3)(2, 4)
    (1, 2)(3, 4)
> C2:=Centralizer(A5,A5!(1,2,3)); C2;
Permutation group C2 acting on a set of cardinality 5
Order = 3
    (1, 2, 3)
> C3:=Centralizer(A5,A5!(1,2,3,4,5)); C3;
Permutation group C3 acting on a set of cardinality 5
Order = 5
    (1, 2, 3, 4, 5)
> C4:=Centralizer(A5,A5!(1,3,4,5,2)); C4;
Permutation group C4 acting on a set of cardinality 5
Order = 5
    (1, 3, 4, 5, 2)
```

MAGMA returns the groups, at least the groups in MAGMA's notation, and their sizes $(4, 3, 5, 5$, resp.). Another way is to type

```
 > S:=SetToIndexedSet(Set(A5_classes));
 > cent_a5:={ Centralizer(A5,S[i][3]) : i in [1..#S]}
 > cent_a5;
{
    Permutation group acting on a set of cardinality 5
    Order = 5
        (1, 3, 4, 5, 2),
    Permutation group acting on a set of cardinality 5
    Order = 60 = 2^2 * 3 * 5
        (1, 2)(4, 5)
        (2, 3)(4, 5)
        (3, 4, 5),
    Permutation group acting on a set of cardinality 5
    Order = 5
        (1, 2, 3, 4, 5),
    Permutation group acting on a set of cardinality 5
    Order = 4 = 2^2
        (1, 3)(2, 4)
        (1, 2)(3, 4),
    Permutation group acting on a set of cardinality 5
    Order = 3
```

```
        (1, 2, 3)
}
```

Up to conjugation, there are no other centralizers in A_5. To find the elements in C_2 for example, type

`> Set(C2);`

or `> SetToIndexedSet(Set(cent_a5))[5];` (note: MAGMA has reordered elements so that C_2 corresponds to the $5th$ element in the set `cent_a5`). MAGMA will return to the first command something like

```
{(1, 3, 2),    (1, 2, 3),    Id(C2)}
```

To the second command, MAGMA will return

```
Permutation group acting on a set of cardinality 5
Order = 3
    (1, 2, 3)
```

Next, we compute their normalizers in A_5:

```
> N1:=Normalizer(A5,C1); N1;
Permutation group N1 acting on a set of cardinality 5
Order = 12 = 2^2 * 3
    (1, 3)(2, 4)
    (1, 2)(3, 4)
    (2, 4, 3)
> N2:=Normalizer(A5,C2); N2;
Permutation group N2 acting on a set of cardinality 5
Order = 6 = 2 * 3
    (1, 2, 3)
    (2, 3)(4, 5)
> N3:=Normalizer(A5,C3); N3;
Permutation group N3 acting on a set of cardinality 5
Order = 10 = 2 * 5
    (1, 2, 3, 4, 5)
    (2, 5)(3, 4)
> N4:=Normalizer(A5,C4); N4;
Permutation group N4 acting on a set of cardinality 5
```

```
Order = 10 = 2 * 5
    (1, 3, 4, 5, 2)
    (2, 3)(4, 5)
```

By typing

`> IsAbelian(C1);`,

to which MAGMA replies `true`, we find out that C_1 is abelian. (It a group of size 4, so of course it is abelian!) Similarly, we find that all the centralizers are abelian (so A_5 is a commutative transitive group) and all the normalizers are non-abelian.

Now, it turns out that any commutative transitive group G has the following property: if $C, C' \subset G$ are any two centralizers then either $C \cap C' = \{1\}$ (i.e., are "disjoint") or $C = C'$. To check this property for $G = A_5$ and $C = C_1$, it suffices to type

```
> [#(C1 meet C1^(a5[i])) : i in [1..#a5]];
[ 1, 4, 4, 1, 1, 1, 1, 4, 1, 1, 1, 4, 1, 1, 1, 1, 1, 1, 1, 1, 1, 4, 1, 1, 1, 1,
1, 1, 1, 1, 1, 4, 1, 1, 1, 1, 1, 1, 1, 4, 4, 1, 1, 1, 1, 4, 4, 1, 1, 1, 1, 1, 1,
1, 1, 1, 4, 1, 1, 4 ]
> [#(C1 meet C2^(a5[i])) : i in [1..#a5]];
[ 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1,
1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1,
1, 1, 1, 1, 1, 1, 1 ]
> [#(C1 meet C3^(a5[i])) : i in [1..#a5]];
[ 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1,
1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1,
1, 1, 1, 1, 1, 1, 1 ]
> [#(C1 meet C4^(a5[i])) : i in [1..#a5]];
[ 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1,
1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1,
1, 1, 1, 1, 1, 1, 1 ]
```

All the elements of each list L end up being either 1 (if their intersection is trivial) or 4 (if they are equal). For example, to see which x satisfies $|C_1 \cap C_1^x| = 4$, type

```
C11_big:=[a5[i] : i in [1..#a5] | 1 lt C11[i]];
```

Note that the output is precisely the elements of N_1, as expected (`Set(N1) eq Set(C11_big);` returns `true`).

Type

```
> RightTransversal(N1,C1);
{@
```

```
    Id(N1),
    (2, 4, 3),
    (2, 3, 4)
@}
```

to get the coset representatives of

$$C_1 = \{(), (1,2)(3,4), (1,3)(2,4), (1,4)(2,3)\}$$

in

$$\begin{aligned} N_1 = \{&(), (2,3,4), (2,4,3), (1,2)(3,4), (1,2,3), (1,2,4), (1,3,2), \\ &(1,3,4), (1,3)(2,4), (1,4,2), (1,4,3), (1,4)(2,3)\}. \end{aligned}$$

Thus the representatives of N_1/C_1 are $\{1, (2,3,4), (2,4,3)\}$ and their cosets are:

```
> r1:=[C1*n : n in SetToIndexedSet(Set(RightTransversal(N1,C1)))];
> r1;
[ RightCoset(GrpPerm: C1, Degree 5, Order 2^2 * (2, 3, 4)), RightCoset(GrpPerm: C1, Degree 5,
Order 2^2 * Id(N1)), RightCoset(GrpPerm: C1, Degree 5, Order 2^2 * (2, 4, 3)) ]
```

Remark 1. If instead you type

```
> CosetTable(N1,C1);
```

then MAGMA returns

```
Mapping from: Cartesian Product<{ 1 .. 3 }, GrpPerm: N1, Degree 5, Order 2^2 * 3> to { 1 .. 3 }
    $1  $2  $3 -$3
1.   1   1   2   3
2.   2   2   3   1
3.   3   3   1   2
```

What does this mean? Recall N_1 has generators $(1,2)(3,4), (1,3)(2,4), (2,4,3)$. (If you don't believe me, type

```
> Generators(N1);
```

which will get MAGMA to tell you this.) For each generator and its inverse, MAGMA computes its effect on the cosets N_1/C_1 as a permutation. There are $3 = |N_1/C_1|$ cosets, so the permutation is in S_3, in this case. In particular, we find that the `CosetAction` command tells us that the permutation representation of N_1 acting on N_1/C_1 is isomorphic to a cyclic subgroup of S_3:

```
> CosetAction(N1,C1);
Mapping from: GrpPerm: N1 to GrpPerm: $, Degree 3, Order 3
Permutation group acting on a set of cardinality 3
```

```
Order = 3
    Id($)
    Id($)
    (1, 2, 3)
Permutation group acting on a set of cardinality 5
Order = 4 = 2^2
    (1, 3)(2, 4)
    (1, 2)(3, 4)
```

3 Representations of A_5

Now we turn to representation theory. Since the centralizers are abelian in this case, all its irreducible representations are 1-dimensional. For the set of all irreducible representations of C_1, you may type either

```
> C1_repns:=CharacterTable(C1);
```

(which returns the irreducible characters of the group), or

```
> C1_repns:=LinearCharacters(C1);
```

(which returns all the 1-dimensional irreducible characters of the group, which is all of them since C_1 is abelian). You can see the output (which is MAGMA's notation for a character of C_1) is the same in either case. Let us abbreviate the characters of C_1 by

$$\mu_{1,1}\ (=1),\quad \mu_{1,2},\quad \mu_{1,3},\quad \mu_{1,4}.$$

The set of irreducible characters of C_1 is denoted C_1^*.

We obtain

```
> C1_repns:=CharacterTable(C1); C1_repns;


Character Table of Group C1
---------------------------



-----------------
Class |   1 2 3 4
Size  |   1 1 1 1
```

```
Order |   1 2 2 2
-----------------
p  =  2   1 1 1 1
-----------------
X.1   +   1 1 1 1
X.2   +   1 1-1-1
X.3   +   1-1 1-1
X.4   +   1-1-1 1

> [C1_repns[i] : i in [1..4]];
[ ( 1, 1, 1, 1 ), ( 1, 1, -1, -1 ), ( 1, -1, 1, -1 ), ( 1, -1, -1, 1 ) ]
```

What does this mean? First, we find the conjugacy classes of each group $C_1, ..., C_4$. Since each of these groups is abelian, the conjugacy classes correspond to the elements of the groups themselves. Consider for example the character `C1_repns[2];` of C_1 which we shall denote by μ. This is a homomophism $\mu : C_1 \to \mathbb{C}^\times$. The vector entry `( 1, 1, -1, -1 )` indicates the values of this character μ on the classes of C_1. For example, $\mu((1,2)(3,4)) = -1$.

Characters of the other centralizers may be determined similarly using MAGMA.

```
> C2_repns:=CharacterTable(C2); C2_repns;


Character Table of Group C2
---------------------------



-------------------
Class |   1   2   3
Size  |   1   1   1
Order |   1   3   3
-------------------
p  =  3   1   1   1
-------------------
X.1   +   1   1   1
```

```
X.2   0   1   J-1-J
X.3   0   1-1-J   J


Explanation of Symbols:
-----------------------

J = RootOfUnity(3)


> C3_repns:=CharacterTable(C3); C3_repns;


Character Table of Group C3
---------------------------



-------------------------------
Class |   1    2    3    4    5
Size  |   1    1    1    1    1
Order |   1    5    5    5    5
-------------------------------
p  =  5   1    1    1    1    1
-------------------------------
X.1   +   1    1    1    1    1
X.2   0   1   Z1 Z1#2 Z1#3 Z1#4
X.3   0   1 Z1#2 Z1#4   Z1 Z1#3
X.4   0   1 Z1#3   Z1 Z1#4 Z1#2
X.5   0   1 Z1#4 Z1#3 Z1#2   Z1


Explanation of Symbols:
-----------------------

# denotes algebraic conjugation, that is,
#k indicates replacing the root of unity w by w^k

Z1     = zeta_5 where zeta_5 is RootOfUnity(5)
```

```
> C4_repns:=CharacterTable(C4); C4_repns;


Character Table of Group C4
---------------------------



-------------------------------
Class |   1    2    3    4    5
Size  |   1    1    1    1    1
Order |   1    5    5    5    5
-------------------------------
p  =  5   1    1    1    1    1
-------------------------------
X.1   +   1    1    1    1    1
X.2   0   1   Z1 Z1#2 Z1#3 Z1#4
X.3   0   1 Z1#2 Z1#4   Z1 Z1#3
X.4   0   1 Z1#3   Z1 Z1#4 Z1#2
X.5   0   1 Z1#4 Z1#3 Z1#2   Z1


Explanation of Symbols:
-----------------------

# denotes algebraic conjugation, that is,
#k indicates replacing the root of unity w by w^k

Z1     = zeta_5 where zeta_5 is RootOfUnity(5)
```

A question which arises later (in the next section) when we induce these characters to A_5 is the following. Is the character

`C1_repns[2]`

of C_1, call it μ, invariant under the action of the normalizer? In fact, since N_1/C_1 is cyclic we find that $\mu^{(2,4,3)} \neq \mu$, so the stabilizer of μ in N_1/C_1 is trivial. We call such a character μ "regular".

Let us abbreviate the characters of C_2 by

$$\mu_{2,1}\ (=1), \quad \mu_{2,2}, \quad \mu_{2,3},$$

the characters of C_3 by

$$\mu_{3,1}\ (=1), \quad \mu_{3,2}, \quad \mu_{3,3}, \quad \mu_{3,4}, \quad \mu_{3,5},$$

and the characters of C_4 by

$$\mu_{4,1}\ (=1), \quad \mu_{4,2}, \quad \mu_{4,3}, \quad \mu_{4,4}, \quad \mu_{4,5},$$

3.1 Character values for A_5

The representations of A_5 are obtained by typing

```
A5_repns:=CharacterTable(A5);
```

MAGMA will reply

```
Character Table of Group A5
---------------------------



---------------------------
Class |   1  2  3    4    5
Size  |   1 15 20   12   12
Order |   1  2  3    5    5
---------------------------
p  =  2   1  1  3    5    4
p  =  3   1  2  1    5    4
p  =  5   1  2  3    1    1
---------------------------
X.1   +   1  1  1    1    1
X.2   +   3 -1  0   Z1 Z1#2
X.3   +   3 -1  0 Z1#2   Z1
X.4   +   4  0  1   -1   -1
X.5   +   5  1 -1    0    0
```

```
Explanation of Symbols:
-----------------------

# denotes algebraic conjugation, that is,
#k indicates replacing the root of unity w by w^k

Z1     = -zeta_5^3 - zeta_5^2 where zeta_5 is RootOfUnity(5)
```

Let us abbreviate them by

$$\pi_1\ (=1), \quad \pi_2, \quad \pi_3, \quad \pi_4, \quad \pi_5.$$

The set of irreducible characters of A_5 is denoted A_5^*.

The notation of the conjugacy classes agrees with that of the ATLAS [Atlas]. In particular, the class of the identity element is the first one; thus the degree of a character is the character value in column 1. The other values depend on the ordering of the conjugacy classes (in `A5_classes`) and on the ordering of the irreducible characters (in `A5_repns`).

The individual values of the characters are obtained using MAGMA's built in evaluation function. The command `A5_repns[i](A5!x)` returns the value of tr $\pi_i(x)$ at the permutation x. For example,

```
> A5_repns[4](A5!(1,2)(3,5));
0
> A5_repns[5](Id(A5));
5
> A5_repns[4](A5!(1,2,3));
1
> A5_repns[4](A5!(1,2,3,4,5));
-1
```

3.2 Induction

Now, let us induce the first (trivial) character of C_1 from C_1 to A_5:

```
> ind1:=Induction(C1_chars[1],A5);
```

To study reducibility of representations [3] using MAGMA, define the **(Schur) scalar product** of a class function χ with a class function ψ on a finite group G by

$$(\chi, \psi) = \frac{1}{|G|} \sum_{g \in G} \chi(g)\psi(g^{-1}).$$

Recall, the scalar product of a character (i.e., the trace of a possibly reducible representation of G) with itself is 1 if and only if it is irreducible.

Is this induced representation irreducible? Type either

```
> IsIrreducible(ind1);
```

(to which MAGMA returns `false`) or

```
> InnerProduct(ind1,ind1);
```

(to which MAGMA returns 6). So, we now know $ind_{C_1}^{A_5} 1$ is not irreducible. What is its decomposition into irreducibles?

$$ind_{C_1}^{A_5} 1 \cong \sum_{\pi \in A_5^*} m(\pi)\pi, \quad m(\pi) \in \mathbb{Z}.$$

To determine the multiplicities $m(\pi)$ it suffices (thanks to Schur orthogonality) to type

```
m:=[InnerProduct(ind1,A5_repns[i]) : i in [1..#A5_repns]];
```

MAGMA returns `[ 1, 0, 0, 1, 2 ]`, so, in the notation of 3.1,

$$ind_{C_1}^{A_5} \mu_{1,1} = ind_{C_1}^{A_5} 1 \cong \pi_1 + \pi_4 + 2\pi_5.$$

Likewise, by typing

```
> ind2:=Induction(C1_repns[2],A5);
> InnerProduct(ind2,ind2);
4
> ind3:=Induction(C1_repns[3],A5);
> InnerProduct(ind3,ind3);
4
> ind4:=Induction(C1_repns[4],A5);
> InnerProduct(ind4,ind4);
4
```

[3] More precisely, reducibility of characters of representations.

we find that none of the induced representations $ind_{C_1}^{A_5}\mu$ are irreducible. Indeed, MAGMA gives

$$ind_{C_1}^{A_5}\mu_{1,2} \cong ind_{C_1}^{A_5}\mu_{1,3} \cong ind_{C_1}^{A_5}\mu_{1,4} \cong \pi_2 + \pi_3 + \pi_4 + \pi_5.$$

Exercise: By typing `ind1:=Induction(C2_chars[1],A5);`, ... show that none of the induced representations $ind_{C_2}^{A_5}\mu$ are irreducible.
Answer: By entering commands similar to those explained above, MAGMA tells us that

$$ind_{C_2}^{A_5}\mu_{2,1} = ind_{C_2}^{A_5}1 \cong \pi_1 + \pi_2 + \pi_3 + 2\pi_4 + \pi_5,$$

$$ind_{C_2}^{A_5}\mu_{2,2} \cong ind_{C_2}^{A_5}\mu_{2,3} \cong \pi_2 + \pi_3 + \pi_4 + 2\pi_5.$$

Exercise: Show that none of the induced representations $ind_{C_3}^{A_5}\mu$ or $ind_{C_4}^{A_5}\mu$ are irreducible either.
Answer: By entering commands similar to those explained above, MAGMA tells us that

$$ind_{C_3}^{A_5}\mu_{3,1} = ind_{C_3}^{A_5}1 \cong ind_{C_4}^{A_5}\mu_{4,1} = ind_{C_4}^{A_5}1 \cong \pi_1 + \pi_2 + \pi_3 + \pi_5,$$

$$ind_{C_3}^{A_5}\mu_{3,2} = ind_{C_3}^{A_5}\mu_{3,5} \cong ind_{C_4}^{A_5}\mu_{4,3} = ind_{C_4}^{A_5}\mu_{4,4} \cong \pi_2 + \pi_4 + \pi_5,$$

$$ind_{C_3}^{A_5}\mu_{3,3} = ind_{C_3}^{A_5}\mu_{3,4} \cong ind_{C_4}^{A_5}\mu_{4,2} = ind_{C_4}^{A_5}\mu_{4,5} \cong \pi_3 + \pi_4 + \pi_5.$$

Another way to determine whether or not these induced representations (or any other ones you might happen to run across) are irreducible is simply to compare their values with those given by the character table of A_5 as given in §3.1. The Frobenius formula for the character of an induced representation in this case becomes particularly simple.

Lemma 1. *Let G be a finite commutative transitive group, C a centralizer in G, $\mu \in C^*$ a character, and let $\pi = ind_C^G\mu$. Then*

$$\mathrm{tr}\,\pi(g) = \begin{cases} \frac{|G|}{|C|}, & g = 1 \\ \frac{1}{|C|}\sum_{n\in N_G(C)} \mu(n^{-1}cn), & c \in Conj(g,G) \cap C \neq \emptyset,\ g \neq 1 \\ 0, & Conj(g,G) \cap C = \emptyset,\ g \neq 1. \end{cases}$$

4 Orbital integrals

To compute orbital integrals in MAGMA, you must write a program (called a "function" in MAGMA; a "procedure" in MAGMA does not return a value).

```
> function AddList(L)
function> total:=0;
function> for i in [1..#L] do
function|for> total:=total+L[i];
function|for> end for;
function> return total;
function> end function;

> function orbital_integral(g,f)
function> vals:=[f(x^(-1)*g*x) : x in a5];
function> return AddList(vals)/#a5;
function> end function;
```

Here f is an as yet undefined function on A_5. First, type

```
> function class_fcn_A5(x,y)
function> if x in Class(A5,y) then
function|if> return 1;
function|if> end if;
function> return 0;
function> end function;
```

Now type

```
> function f(x)
function> return class_fcn_A5(x,A5!(1,2)(3,5))+3*class_fcn_A5(x,A5!(1,2,3))-cl\
ass_fcn_A5(x,Id(A5));
function> end function;
```

The orbital integral of this function on A_5 is computed by typing

```
> orbital_integral(A5!(1,2,3),f);
```

to which MAGMA responds 3.

5 Principal series of $SL(2,4)$

Here we wish to use MAGMA to examine which (if any) characters of the standard Borel subgroup $B = \{\begin{pmatrix} * & * \\ 0 & * \end{pmatrix}$ in $G = SL(2,4)$ induce irreducibly to G.

As was indicated already, we may enter $SL(2,4)$, as a matrix group, into MAGMA by typing

```
> SL24:=SL(2,4);
```

but then we would have the problem of determining which subgroup of G the Borel is. To find out which group is the Borel subgroup of this matrix group, let us first list all the entries whose lower left corner entry is a 0:

```
> S:=SetToIndexedSet(Set(SL24));
> L0:=[ i : i in [1..60] | S[i][2,1] eq 0 ];
> Borel:=[S[i] : i in L0];
> B:=MatrixGroup< 2, GF(4) | Borel>;
> L00:=[ i : i in [1..60] | (S[i][1,2] eq 0) and (S[i][2,1] eq 0)];
> Diag:=[S[i] : i in L00];
> A:=MatrixGroup< 2, GF(4) | Diag>;
```

5.1 Representations of $SL(2,4)$

The irreducible representations of B and G are obtained by typing

```
> A_repns:=CharacterTable(A);
> B_repns:=CharacterTable(B);
> SL24_repns:=CharacterTable(SL24);
```

We abbreviate these irreducible representations of B by

$$\psi_1\ (=1),\quad \psi_2,\quad \psi_3,\quad \psi_4.$$

Note ψ_1, ψ_2, ψ_3 are characters of A extended trivially to B. We may abbreviate these irreducible representations of G by

$$\pi'_1\ (=1),\quad \pi'_2,\quad \pi'_3,\quad \pi'_4,\quad \pi'_5.$$

To relate these representations to those obtained in the previous section, we look at the conjugacy classes of G.

```
 > A_classes:=Classes(A);
 > B_classes:=ConjugacyClasses(B);
 > SL24_classes:=Classes(SL24);
```

To get class representatives, type

```
> a_classes:=SetToIndexedSet(Set(A_classes));
> a_class_reps:=[a_classes[i][3] : i in [1..#a_classes]];
> a_class_reps;
[
    [  $.1      0]
    [    0 $.1^2],

    [$.1^2      0]
    [    0   $.1],

    [    1      0]
    [    0      1]
]
```

Exercise: Obtain class representatives for B and $SL(2,4)$.
Answer:

```
> b_class_reps;
[
    [    1 $.1^2]
    [    0      1],

    [$.1^2      0]
    [    0   $.1],

    [  $.1      0]
    [    0 $.1^2],

    [    1      0]
    [    0      1]
]
> sl24_class_reps;
[
```

```
    [    1     0]
    [  $.1     1],

    [    1     1]
    [    1     0],

    [  $.1 $.1^2]
    [  $.1     0],

    [  $.1     1]
    [$.1^2     1],

    [    1     0]
    [    0     1]
]
```

By comparing the character values of these π_i''s with those in the previous section, we find that under the above correspondence of conjugacy classes, π_i matches with π_i' in the sense that their characters are equal on the corresponding classes, $i = 1, 2, 3, 4, 5$.

5.2 Induced representations of $SL(2, 4)$

The induced representations from B to G are obtained by typing

```
> rho1:=Induction(B_repns[1],G);
> rho2:=Induction(B_repns[2],SL24);
> rho3:=Induction(B_repns[3],SL24);
> rho4:=Induction(B_repns[4],SL24);
```

To determine their reducibility, type

```
> m1:=[InnerProduct(rho1,x) : x in SL24_repns]; m1;
[ 1, 0, 0, 1, 0 ]
> m2:=[InnerProduct(rho2,x) : x in SL24_repns]; m2;
[ 0, 0, 0, 0, 1 ]
> m3:=[InnerProduct(rho2,x) : x in SL24_repns]; m3;
```

```
[ 0, 0, 0, 0, 1 ]
> m4:=[InnerProduct(rho2,x) : x in SL24_repns]; m4;
[ 0, 0, 0, 0, 1 ]
```

This tells us that $ind_B^G\ 1$ is reducible. The irreducible representations $ind_B^G\ \psi_2$, $ind_B^G\ \psi_3$ are the "principal series" representations. It also tells us that the induced representation $ind_B^G\ \psi_4$ is reducible and vanishes on the "regular hyperbolic set" $A - \{1\}$.

5.3 Principal series, revisited

We have seen all the principal series representations but their construction wasn't the usual one. Usually, on starts with a character of A then extends to B and induces. MAGMA can do this.

```
> L01:=[ i : i in [1..60] | (S[i][1,1] eq 1) and (S[i][2,1] eq 0)];
> Nil:=[S[i] : i in L01];
> N:=MatrixGroup< 2, GF(4) | Nil>;
> A0,f:=quo<B | N>;
> f;
Mapping from: GrpMat: B to GrpPerm: A0
> A0;
Permutation group A0 acting on a set of cardinality 3
    (1, 2, 3)
    (1, 3, 2)
    (1, 2, 3)
    Id(A0)
    (1, 3, 2)
    (1, 2, 3)
    (1, 3, 2)
    Id(A0)
    (1, 2, 3)
    (1, 3, 2)
    Id(A0)
> #A0;
3
> A0_repns:=CharacterTable(A0);
```

```
> B_rep1:=LiftCharacter(A0_repns[1],f,B);
> B_rep2:=LiftCharacter(A0_repns[2],f,B);
> B_rep3:=LiftCharacter(A0_repns[3],f,B);
> ind1:=Induction(B_rep1,SL24);
> ind2:=Induction(B_rep2,SL24);
> ind3:=Induction(B_rep3,SL24);
> m1:=[InnerProduct(ind1,x) : x in SL24_repns]; m1;
[ 1, 0, 0, 1, 0 ]
> m2:=[InnerProduct(ind2,x) : x in SL24_repns]; m2;
[ 0, 0, 0, 0, 1 ]
> m3:=[InnerProduct(ind3,x) : x in SL24_repns]; m3;
[ 0, 0, 0, 0, 1 ]
```

6 Construction of $PSL(2,5)$ as a quotient

We shall first work with $SL(2,5)$ then mod out by its center.

Type

```
> SL25:=SL(2,5);
```

and then type

```
> Z_SL25:=Center(SL25);
```

to enter its center. As we know, $PSL(2,5) \cong A_5$.

How do you enter $SL(2,5)/Z$ (or any quotient group, for that matter) into MAGMA? It is easy if you know the right command. First, you must have a group and a normal subgroup (actually, MAGMA allows more general constructions which we won't need here). The quotient group $PSL(2,5)$ is simply given by

```
> G:=SL25/Z_SL25;
```

It is just that easy. The elements are permutations.

7 A_5 as a finitely presented group

The group A_5 has finite presentation [Atlas]

$$< \quad a, b \mid a^2 = b^3 = (ab)^5 = 1 \quad > .$$

To enter this into MAGMA, type

```
> F2<a,b> := FreeGroup(2);
> rels:={a^2=Id(F2), b^3=Id(F2), (a*b)^5=Id(F2)};
> G:=quo<F2|rels>;
```

To reinterprete this as a permutation group, type:

```
> T:=CosetTable(G, H);
> G0:=CosetTableToPermutationGroup(G,T);
> G0;
Permutation group G0 acting on a set of cardinality 60
    (1, 2)(3, 7)(4, 8)(5, 9)(6, 10)(11, 19)(12, 20)(13, 21)(14, 22)(15, 23)(16, 24)(17, 25)(18,
        26)(27, 37)(28, 38)(29, 30)(31, 39)(32, 40)(33, 41)(34, 35)(36, 42)(43, 54)(44, 45)(46,
        55)(47, 56)(48, 49)(50, 51)(52, 57)(53, 58)(59, 60)
    (1, 3, 4)(2, 5, 6)(7, 11, 12)(8, 13, 14)(9, 15, 16)(10, 17, 18)(19, 26, 27)(20, 28, 29)(21, 30,
        31)(22, 32, 23)(24, 33, 34)(25, 35, 36)(37, 43, 44)(38, 45, 46)(39, 47, 48)(40, 49, 50)(41,
        51, 52)(42, 53, 54)(55, 59, 56)(57, 60, 58)
> #G0;
60
> IsSimple(G0);
true
```

This brief paper has only given an overview of some of the most basic features of MAGMA. For more, see the documentation [MAGMA].

Acknowledgements: I thank John Cannon for several comments and suggestions for improvements.

References

[Atlas] Robert Wilson, Peter Walsh, Jonathan Tripp, Ibrahim Suleiman, Stephen Rogers, Richard Parker, Simon Norton, Steve Linton and John Bray, **ATLAS of Finite Group Representations**, at `http://www.mat.bham.ac.uk/atlas/html/A5.html`

[MAGMA] W. Bosma, J. Cannon, C. Playoust, "The MAGMA algebra system, I: The user language," J. Symb. Comp., 24(1997)235-265. (See also the MAGMA homepage at `http://www.maths.usyd.edu.au:8000/u/magma/`)

[R] J. J. Rotman, **An introduction to the theory of groups**, 4th ed, Springer-Verlag, Grad Texts in Math 148, 1995